


* Chapter 1. Hspice

IC CAD 실험 Analog part

* Hspice simulation

TR level circuit simulator! – text 기반



* Hspice simulation


Example> Inverter 설계 및 simulation

```
mkdir hspice
cd hspice
cp ../../hspice/netlist.sp .
cp ../../hspice/PMOS_VTL.inc .
cp ../../hspice/NMOS_VTL.inc .
ls

vi netlist.sp
```

* Hspice simulation

Example> Inverter 설계 및 simulation



Netlist

```
.subckt Inverter out in vdd vss  
M_p out in vdd vdd PMOS_VTL w=10u l=0.05u  
M_n out in vss vss NMOS_VTL w=5u l=0.05u  
.ends
```

.subckt **Inverter** **out** **in** **vdd** **vss**


회로 만들기

회로 이름

회로 바깥에서 보이는 port 들

* Hspice simulation

Example> Inverter 설계 및 simulation



Netlist

```
.subckt Inverter out in vdd vss
M_p out in vdd vdd PMOS_VTL w=10u l=0.05u
M_n out in vss vss NMOS_VTL w=5u l=0.05u
.ends
```

Mosfet Drain 의 Gate 의 Source 의 Bulk 의
의 이름 node 명 node 명 node 명 node 명



Mosfet 의 type

Mosfet 의 width 와 length

Inverter 설계 끝!

* Hspice simulation

Example> Inverter 설계 및 simulation

```
@ hspice directory
```

```
vi sim_inverter.sp
```

[Vi editor 기본 단축키]

Insert : insert/replace

Esc: 커서모드로 전환


:q! 저장하지 않고 끝내기

:wq! 저장하고 끝내기

* Hspice simulation

Example> Inverter 설계 및 simulation

DC simulation



Vin 이라는 voltage source 를 0.001 단위로 0에서 1까지 변화시키면서 각각의 node 들의 값이 어떻게 되는지 simulation 해라.

****(첫줄은 자동 주석 처리)

```
.include 'PMOS_VTL.inc'
.include 'NMOS_VTL.inc'
.include 'netlist.sp'
```

Model 및
만들어 놓
은 회로 불
러오기

```
.OPTIONS POST NODE LIST
```

```
V1 vdd 0 1
V2 vss 0 0
Vin in vss 0.5
```

Voltage 정의
해 주기

Inverter 라는 회로 불러오기

```
x_inv out in vdd vss inverter
```

```
c_out out vss 1p
```

Capcitor 정의

```
.dc Vin 0 1 0.001
.END
```

* Hspice simulation

Example> Inverter 설계 및 simulation



```
hspice inv_sim.sp > a.lis &
```

```
>info : **** hspice job concluded
```

```
wv &
```



* Hspice simulation

Example> Inverter 설계 및 simulation
파형 확인하기



* Hspice simulation


Example> Inverter 설계 및 simulation



* Hspice simulation

Example> Inverter 설계 및 simulation

AC simulation




Ac sweep 을 frequency 축을 decade 단위로 하여, 각 decade 마다 1000개의 data 를 찍고, 100KHz 에서 부터 10GHz 까지의 주파수 응답을 simulation 해라

```
*****  
KHz : k  
MHz : meg or x  
GHz : g  
.include 'PMOS_VTL.inc'  
.include 'NMOS_VTL.inc'  
.include 'netlist.sp'  
  
.OPTIONS POST NODE LIST  
  
V1 vdd 0 1  
V2 vss 0 0  
Vin in vss 0.44 ac=1  
  
x_inv out in vdd vss Inverter  
c_out out vss 1p  
  
.ac dec 1000 100k 10g  
.END
```


* Hspice simulation

Example> Inverter 설계 및 simulation



* Hspice simulation

Example> Inverter 설계 및 simulation
transient simulation



Transient sweep 을 1ps 마다 data
를 찍어서 100ns 까지 simulation
해라

```
.include 'PMOS_VTL.inc'  
.include 'NMOS_VTL.inc'  
.include 'netlist.sp'
```

```
.OPTIONS POST NODE LIST
```

```
V1 vdd 0 1
```

```
V2 vss 0 0
```

```
Vin in vss sin(0.5 0.2 100x)
```

0.5 V ± 0.2V 의 swing 을 가지는

100MHz 의 sin 파 신호 생성

```
x_inv out in vdd vss Inverter
```


```
c_out out vss 1p
```

```
.tran 1p 100n
```

```
.END
```


* Hspice simulation

Example> Inverter 설계 및 simulation



* 실습

차동 증폭기 (differential amplifier) 의 설계 및 시뮬레이션



DC, AC, Transient sweep 후 파형 확인

Differential input signal generator

| | | | |
|-------|--------|------|-------------|
| Vin | in | vss | ac=1 |
| Voff | cm | vss | 0.75 |
| Einp | inp cm | VCVS | in vss 0.5 |
| Einn | inn cm | VCVS | in vss -0.5 |
| Vbias | bias | vss | 0.7 |

```
.dc vin -0.5 0.5 0.001
.ac dec 1000 100k 10g
.tran 1p 100n
```

모든 NMOS 의 bulk 는 vss 로
모든 PMOS 의 bulk 는 vdd 로 연결!

* Appendix

• Parameter 설정 및 parameter sweep

Hspice netlist 에서 parameter 설정은 다음과 같이 한다. 예를 들어 어떤 dc voltage 값을 cont 라는 parameter (변수) 값으로 지정하고 싶다면,

```
Vcont cont vss 'cont'
```

와 같이 ‘ ‘ 안에 변수 값을 써 주면 된다. 그리고

```
.param cont=0.6
```

과 같이 선언해주면 cont 에는 0.6 이란 값이 들어가게 된다.

‘cont’ 값을 바꿔가며 parameteric sweep 을 하려면,

```
.dc vx 0 1 0.001 sweep cont 0 1 0.1
```

과 같이 설정해준다. 이렇게 해 주면, cont 값을 0 에서 1 까지 0.1 씩 변화시키며 10번 반복하여 dc sweep 을 하게 된다.

이는 ac sweep, transient sweep 에서도 마찬가지로 적용할 수 있다.

```
.ac dec 1000 100k 10g sweep cont 0 1 0.1
```

```
.tran 1p 100n sweep cont 0 1 0.1
```